

FIRST EDITION

SPICE UP YOUR SHEET LIFE

35 unusual tips & tricks for
Excel & Google Sheets



by benlcollins.com

Contents.

INTRODUCTION

SHORTCUTS

1. Common keyboard shortcuts in Excel

FORMATTING

2. Conditional formatting of a whole row based on single column
3. Use the CELL() function to find out information about a specific cell
4. Quickly strip out the date portion of a timestamp
5. Converting months from text to numbers (e.g. November → 11)
6. Quickly create column headings
7. Format data by perform mathematical operations on a range with Paste Special

LOOKUPS

8. VLOOKUP() with the TRUE parameter for approximate matching
9. Leftwards VLOOKUP() using the CHOOSE() function
10. Lookup with multiple search criteria
11. SUMIF with OR conditions

MANIPULATING DATA

12. Use the FILTER() function on a range to return only the values you want
13. Use the RANK() function to find the order of a value in a list
14. UNIQUE() function in Google Sheets
15. Writing code in your spreadsheets with the QUERY() function
16. Database functions

CHARTS

17. Create charts with the REPT() function
18. Create mini charts inside a cell with Sparklines
19. Motion chart in Google Sheets
20. Google EXPLORE function
21. Creating a histogram with the FREQUENCY() function

GOOGLE SPECIFIC

22. Web scraping with the IMPORTHTML() and IMPORTXML() functions
23. Tag people in comments
24. Detect the language in a given string of text in Google Sheets
25. Build a translation tool in Google Sheets
26. Check for a valid email in Google Sheets
27. Check for a valid URL (web address) in Google Sheets
28. Sourcing financial data in seconds with the GOOGLFINANCE() function
29. Getting data from different Google Sheets

OTHER MISCELLANEOUS

30. Project future results with the GROWTH() function
31. Adding new lines inside formulas to make them more readable
32. Adding funky characters into your worksheets
33. Use CHOOSE and a random number generator to pick from a list
34. Excel intersect operator
35. Bonus - links to other interesting, weird tricks!

CLOSING COMMENTS

Hello.

Did you know that you can build a language translation tool in a Google sheet? Or create a chart of stock prices in under 30 seconds? All possible and covered in these pages! (See pages 58 and 62 for these specific case studies.)

What about other weird tricks like looking up leftwards with VLOOKUP? Or lookups with multiple criteria? Writing SQL code in your spreadsheets? Check, check and check. All possible and covered in this book. See pages 20, 22 and 34 respectively.

This book contains 35 detailed case studies, covering practical and esoteric, easy and difficult, short and long, Excel and Google Sheet tricks and techniques. You may have seen some of them before but my hope is that you'll find some magic in these pages, to inspire you or show you new ways to use spreadsheets you didn't know were possible.

Feedback is welcomed! If you find any errors or feel anything merits further clarification, please don't hesitate to drop me an email ben.collins@gmail.com.

Likewise, if you have your own favorite, unusual or neat way of using spreadsheets, and would like to share, then I'd love to hear from you. My aim is to keep adding to this book, to reach 50, 75, 100 etc. tips and tricks. I'll credit anyone who contributes.

Happy filtering, sorting, summing, pivoting, charting and coding!

Ben Collins

January 2016

#1.

Common Keyboard Shortcuts

Let's start with keyboard shortcuts. I'm including this because it's SO important to master. It's one of the single best investments of time you can make to further your spreadsheet skills.

It's all about reducing your reliance on the mouse and instead harnessing the **awesome efficiency** of navigating spreadsheets from the keyboard.

It's a sure sign of an spreadsheet ninja when they can perform many tasks without needing to touch the mouse.


You may know or use some of these already, but if you don't, it's worth putting in the time to master them. Find out the keystrokes for your common tasks and **practise, practise and practise some more until you're blue in the face**¹. At first it'll be slow and painstaking, but persevere!

In no time you'll be *flying around Excel or Google Sheets* from your keyboard much more quickly, leaving your mouse-wielding colleagues for dust.

Without further ado, here are some of the most useful shortcuts in Excel and Google Sheets:

1. **Quickly move around your data ranges** by holding the **Ctrl key** (or **Cmd key** on a Mac) and the **Left** or **Right** or **Up** or **Down Arrow** keys. **Ctrl** and **Down Arrow** will move to the last continuous row of data in your range. **Ctrl** and **Up Arrow** will take you back to the top. **Ctrl** with the **Left** or **Right Arrow** unsurprisingly take you to the left or right edges of your range. Try this over and over until you get really comfortable. It's the fastest way to *whizz* around your data.

¹ at which stage, you could consider applying to [join the Blue Man Group](#)

2. **Quickly highlight your data** by combining what you've just learned above with the shift key. Hold down **Ctrl key** (or **Cmd key** on a Mac) and the **Shift key** with your left hand and the **Left** or **Right** or **Up** or **Down Arrow** keys with your right hand. This will highlight your data ranges in one easy motion, much easier than using the mouse, especially if you work with large datasets. (Tip: try holding down just the **Shift key** and one of the **Arrow keys**, to highlight specific ranges of data.)
3. **Move quickly between your worksheets (Excel only)** by holding down the **Ctrl key** and **Page Up** or **Page Down**. This is incredibly useful if you have large numbers of worksheets. Work that keyboard to effortlessly cycle through the worksheets.
4. **Toggle between absolute and relative cell references** using the **F4 key** when inside a cell with a reference to another cell. This will become your best friend if you work a lot with lookup formulas, requiring fixed lookup table references.
5. **Highlight a whole row** by holding down the **Shift key** and hitting **Spacebar**. Great for inserting additional lines, formatting titles or moving data around. (Pro tip: combine this with the **Shift** and **Arrow keys**, or **Ctrl** and **Shift** and **Arrow keys** to highlight multiple rows!)
6. **Similarly, highlight a whole column** by holding down the **Ctrl key** and hitting **Spacebar**.
7. **Quickly insert or delete rows or columns (Excel only)** using the **Shift** and **Spacebar**, or **Ctrl** and **Spacebar** mentioned above, you can then easily add or delete whole rows or columns by pressing the **Ctrl key** and the **+ key**.
8. **Bring up the right click menu and easily access those options (Excel only)** by now, you're starting to get the hang of using the keyboard to fly around your workbook, executing at lightning speed. But then you find yourself wanting to right click your mouse....wait, hang on! There's a key for that menu. Look for the menu key  on your keyboard and if you have it, tap it to bring up the right-click menu. If you don't have that key then you hold down the **Shift key** and the **F10 key** to access the right-click menu menu.

9. **Insert filters to your data range (Excel only)** without touching the mouse. With your cursor lying somewhere within your data range, tap the **Alt key** once, then tap the **A key**, then tap the **T key**, to add filters across the top of your range. Make the keystrokes quite deliberate at first and watch the menu at the top of your Excel window as you navigate your way to adding filters using only the keyboard.
10. **Now you've added filters, how'd you access them?** Move your cursor one of the cells with a filter symbol (should be the header row of your data) and then hold down the **Alt key** and press the **Down Arrow key**.
11. **Insert a pivot table faster than you can say Jack Robinson² (Excel only)**, by tapping the **Alt Key** once, then the **N key** and then the **V key**. Again, watch the Excel menu at the top of the window to see what's going on. This will save you aeons of time if you work with pivot tables a lot, allowing you to throw up a quick pivot to test an idea.
12. **Quickly jump into a cell to edit the contents**, by pressing the **F2 key**. Yowser!
13. **Find the source data used in a function**, by pressing **Ctrl** and **[**.
14. **Fill down a whole row**, by using **Ctrl** and **D** on a PC or **Cmd** and **D** on a Mac. This applies the Fill Down command and copies the contents and format of the top cell of the range to those underneath. This slick shortcut will make you look like an Excel magician in front of your colleagues!

² [https://en.wikipedia.org/wiki/Jack_Robinson_\(mythical_person\)](https://en.wikipedia.org/wiki/Jack_Robinson_(mythical_person))

#2.

Conditional formatting of a whole row based on single column

Conditional formatting of a whole row	
TOOL	Excel & Google Sheets
TYPE	Formatting
DIFFICULTY	Medium
OFFICIAL DOCUMENTATION	Excel docs / Google Sheets docs

Here, we're going to apply conditional formatting to a whole row, but use a single cell within that row to determine the formatting. Think of a tracker sheet with a "Completed" column which records "Yes" when a task is completed. Wouldn't it be neat if you could turn the whole row green say? Or strikethrough the text to really emphasize it's completed?

Well, it's possible! But it is slightly tricky. I find myself wanting to use this technique quite often but always forgetting the exact steps, which then entails messing around with the formulas to get right. What a waste of time!


So I've documented the steps to do this in Google Sheets:

1. Highlight the table (or range) of rows that you want to apply the conditional formatting to, e.g. A1:C12
2. Go to menu **Format > Conditional formatting...**
3. This opens up a conditional formatting pane on the right side of your browser window.
4. The range you selected will be displayed in the "**Apply to Range**" section, so leave that set.

- In the “**Format cells if...**” menu option, choose “**Custom formula is**”
- Type the following formula into the empty text box underneath:

= $\$c\$2:\$c\$12<>0$

- This formula will color all rows that have an entry in column D, i.e. where they are non-blank.
- In the “**Formatting style**” section, choose a default style (e.g. green for done, red for problem) or a custom style for your formatting.
- Click “**Done**” and Boom! Custom formatting applied to the whole row. Now whenever you add values into column D, the formatting will automatically be applied to that whole row.



If this embedded video is not playing for you, [please check it out online here.](#)

#3.

Use the CELL() function to find out information about specific cells

Find out information about your cells	
TOOL	Excel & Google Sheets
TYPE	Function
DIFFICULTY	Medium
OFFICIAL DOCUMENTATION	Excel docs / Google Sheets docs

The CELL() function is a handy way of finding out information, or metadata, about the contents, structure and datatype of a cell. For example, you can use it to find the contents of a cell, the address of a cell, the datatype of a cell or the dimensions of a cell.

Let's see it in action with an example. We have this information in cells A1 to C2:

	A	B	C
1	Testing	123	25-Nov-2015
2	Testing 123	Hello!	

Use the following CELL() formulas to retrieve information about specific cells:

Formula	Result	Explanation
<code>=cell("width",A1)</code>	21	Returns the column width of cell A1
<code>=cell("width",C1)</code>	28	Returns the column width of cell C1
<code>=cell("type",A1)</code>	I	Returns "I" to indicate the cell contains plain text

<code>=cell("type",C1)</code>	b	Returns "b" to indicate "blank" cell
<code>=cell("type",B1)</code>	v	Returns "v" to indicate "value" because the cell contains other data that is not text or blank
<code>=cell("row",A2)</code>	2	Returns 2, the row number of the top left cell of the range selected
<code>=cell("address",A2)</code>	\$A\$2	Returns \$A\$2, the absolute reference of the top left cell of the range selected
<code>=cell("contents",A1:C2)</code>	Testing	Returns "Testing", the value contained in the top left cell of the range selected
<code>=cell("contents",B1:D2)</code>	123	Returns "123", the value contained in the top left cell of the range selected

#4.

Quickly extract the date portion of a timestamp

Use the INT() function to extract the date portion of a timestamp

TOOL	Excel & Google Sheets
TYPE	Function
DIFFICULTY	Easy
OFFICIAL DOCUMENTATION	Excel docs / Google Sheets docs

Ever worked with timestamps (a date with a time component) in your spreadsheet and wished you could easily get at just the date portions of timestamps? In other words, easily remove those pesky times because they're too granular for what you need.

Well, there's a quick and simple way to do that with the INT() function.

Dates in spreadsheets are stored as 5 digit integers and timestamps are stored as 5 digit *decimal* numbers with several decimal places. So we can use the INT function to remove the decimal places, leaving just the integer part of the number, corresponding to the date only.

	A	B	C
1	Full Date	12/05/2013 at 01:33:00pm	12/05/2013
2	Full Date formatted as number	41,406.5646	41,406.0000
3			
4			

Neat, huh?

#5.

Converting months from text to numbers (e.g. November → 11)

Converting between word and number dates	
TOOL	Excel & Google Sheets
TYPE	Function
DIFFICULTY	Medium/Hard
OFFICIAL DOCUMENTATION	Excel docs / Google Sheets docs

Turns out it's harder than you'd expect to change the number 6 to the month "June", or 11 to the month "November", or vice versa.

Why would you want to do that? Well, as an example, you might want to use the number (11) in a concatenation formula.

There's no easy function to swap between the two formats but we can still avoid using a nasty lookup-solution involving a separate table.

We can use the CHOOSE function or the awesome INDEX and MATCH formulas to achieve this, and put a static array of months or numbers into the formula. It's a super cool technique, so let's see it in action.

Number to word format:

	A	B
1	11	November

To convert from a number (11) to a month written as a word (November), as shown in the table above, use the CHOOSE formula as follows:

```
=choose(A1, "January", "February", "March", "April", "May", "June", "July",
"August", "September", "October", "November", "December")
```

We can also achieve the result with the following INDEX/MATCH combination formula (in cell B1 above):

```
=index({"January";"February";"March";"April";"May";"June";"July";
"August";"September";"October";"November";"December"},MATCH(A1,{1,2,3,4
,5,6,7,8,9,10,11,12},0),1)
```

Word format to number:

	A	B
1	November	11

To convert from the month written as a word (November) to the number (11), use the following formula (in cell B1 above):

```
=MATCH(A1, {"January";"February";"March";"April";"May";"June";"July";
"August";"September";"October";"November";"December"}, 0)
```

P.S. one other way to convert dates to a month only format, written as a word. Check this out for three letter months (e.g. "Jan"):

```
=text(today(), "MMM")
```

Or this for the full months (e.g. "January"):

```
=text(today(), "MMMM")
```

Replace **today()** with your date.


#6.

Efficiently create column headings

Efficiently create column headings	
TOOL	Excel & Google Sheets
TYPE	Method
DIFFICULTY	Easy
OFFICIAL DOCUMENTATION	N/a

This is a quick and easy trick for filling out column headings:

1. Start by highlighting the range where the column headings will exist.
2. Type the first column heading in the first cell.
3. Hit **Enter** and the cursor will jump sideways, not down, to the adjacent cell, where you can enter your next column heading.
4. Repeat!



If this embedded video is not playing for you, [please check it out online here](#).

#7.

Format data by performing mathematical operations on a range with Paste Special

Perform mathematical operations with Paste Special	
TOOL	Excel
TYPE	Method
DIFFICULTY	Medium
OFFICIAL DOCUMENTATION	Excel docs / Google Sheets docs

This is a super useful trick for quickly performing a mathematical operation on a whole range of numbers.


For example, imagine you have a table full of billion dollar financial numbers - 9 digit numbers - so big in other words, that they clutter up your table and make it hard to read. Wouldn't it be better if you could quickly convert all of them to something more manageable?

For example, what if all the numbers in your table could quickly be converted from this \$2,350,789,112 to this \$2.35? You could then just annotate your table to show that this number represented billions.

Well, it's easy with the following steps:

1. Type the value 1,000,000,000 into a cell outside of the table.
2. Highlight this cell and copy it (**Ctrl + C** on a PC or **Cmd + C** on a Mac).
3. Now highlight the whole range of your table.

4. Go to menu **Home > Paste > Paste Special** (or press **Ctrl + Alt + V** on a PC, or **Cmd + Alt + V** on a Mac).
5. In the Paste Special menu, choose **Divide** under the Operation section.
6. Click OK and see the table change to display the numbers as billions.



If this embedded video is not playing for you, [please check it out online here.](#)

#8.

VLOOKUP with TRUE parameter for approximate matching

Approximate matching with the VLOOKUP() function	
TOOL	Excel & Google Sheets
TYPE	Function
DIFFICULTY	Medium
OFFICIAL DOCUMENTATION	Excel docs / Google Sheets docs

Hands up if you've used the VLOOKUP function before? Yeah, ok, so probably most of you put your hands up. It's one of the most famous spreadsheet functions, and using it elevates you from beginner to intermediate spreadsheet user.

But how many of you have ever set that last parameter (Range Lookup) to TRUE? Be honest now, I'm sure most of you are accustomed to writing FALSE and haven't thought anymore about it. False gives you an exact match for your lookup value and it's what we want most of the time (probably 99% of the time).

Sometimes however, we want an approximate match and the VLOOKUP can handle this with aplomb.

Imagine the scenario where you have a table of different bank interest rates depending on the account balance and you need to use the rate to calculate the interest for a few thousand accounts say. It's going to take you a while to manually add the interest rates. You could use VBA and write a program to do it, but that's overkill.

Instead let's use VLOOKUP with TRUE, as follows:

1. Imagine we have the following table of interest rates in cells A1 to B5 (shown on the left side of screenshot) and we want to apply them to the table of account balances listed in column D (shown on the right side of screenshot).

	A	B	C	D
1	Amount	Interest Rate		Account
2	\$0 - \$49,999	0.50%		\$146,025
3	\$50,000 - \$99,999	0.75%		\$358,617
4	\$100,000 - \$249,999	1.00%		\$409,288
5	\$250,000+	1.25%		\$24,575
6				\$441,400
7				\$393,508
8				\$63,766
9				\$279,009
10				\$383,885
11				\$352,065
12				\$174,735
13				\$492,386
14				\$51,678
15				\$416,106

2. First, create a table of just the lower bounds of the interest rate ranges, in this example in cells A7 to B10.

6			
7		\$0	0.50%
8		\$50,000	0.75%
9		\$100,000	1.00%
10		\$250,000	1.25%
11			
12			

3. Second, in the column adjacent to your list of account balances, add the following formula:

=VLOOKUP(D1, \$A\$7:\$B\$10, 2, TRUE)

4. Voila! There's a quick way to determine all those interest rates.

The final screenshot looks like this:

fx =VLOOKUP(D2,\$A\$7:\$B\$10,2,TRUE)					
	A	B	C	D	E
1	Amount	Interest Rate		Account	
2	\$0 - \$49,999	0.50%		\$146,025	1.00%
3	\$50,000 - \$99,999	0.75%		\$358,617	1.25%
4	\$100,000 - \$249,999	1.00%		\$409,288	1.25%
5	\$250,000+	1.25%		\$24,575	0.50%
6				\$441,400	1.25%
7	\$0	0.50%		\$393,508	1.25%
8	\$50,000	0.75%		\$63,766	0.75%
9	\$100,000	1.00%		\$279,009	1.25%
10	\$250,000	1.25%		\$383,885	1.25%
11				\$352,065	1.25%
12				\$174,735	1.00%
13				\$492,386	1.25%
14				\$51,678	0.75%
15				\$416,106	1.25%

#9.

Leftwards VLOOKUP using the CHOOSE function

Leftwards VLOOKUP() using the CHOOSE() function	
TOOL	Excel
TYPE	Function
DIFFICULTY	Hard
OFFICIAL DOCUMENTATION	Excel docs

VLOOKUP is a popular formula but suffers from a major drawback. You can't lookup data to the left!

(Note: we could use an INDEX/MATCH formula combination to solve this, but it is also possible to do this with a VLOOKUP, and whilst the INDEX/MATCH option is definitely better to use in practice, this VLOOKUP method is cool, fun and interesting, so I'm sharing it here with you.)

So how do we create a leftwards VLOOKUP?

What we do is create a new virtual table with an Array Formula, where the columns are switched, so the VLOOKUP can work as normal on this temporary, virtual table. But there is an additional nuance, you need to enter this as an array formula, by hitting **Ctrl + Shift + Enter** (on a PC) or **Cmd + Shift + Enter** (on a Mac).

So, imagine this is your table:

	A	B	C
1	Profit (m)	Quarter	
2	\$0.87	Company A	
3	\$0.70	Company B	
4	\$1.79	Company C	
5	\$1.36	Company D	
6	\$2.50	Company E	
7			
8			
9			

Want to know the profit for company D using a VLOOKUP?

In **Google Sheets**, try typing this formula into cell E2, and press Ctrl + Shift + Enter (on a PC) or Cmd + Shift + Enter (on a Mac) to ensure it's entered as an Array formula:

```
=ArrayFormula(VLOOKUP(D2, {$B$2:$B$6, $A$2:$A$6}, 2, FALSE))
```

In **Excel**, try this one, again making sure you enter as an Array formula:

```
{=VLOOKUP(D2, CHOOSE({1, 2}, $B$2:$B$6, $A$2:$A$6), 2, FALSE)}
```

Bingo!

It returns \$1.36 as we want.

#10.

Lookup values based on multiple criteria

Multiple criteria lookup	
TOOL	Excel & Google Sheets
TYPE	Function
DIFFICULTY	Hard
OFFICIAL DOCUMENTATION	Excel docs / Google Sheets docs

This technique takes the humble lookup idea a stage further, allowing you to lookup based on multiple criteria.

Normally when performing a lookup (using either VLOOKUP or INDEX/MATCH), you'll have one search term (e.g. "January" or "Company A") that you then go and look up in your main table to retrieve data from. But what if you wanted to lookup January sales for Company A at the same time, i.e. based on both of those search criteria.

Here's a screenshot of an example:

The screenshot shows a Google Sheet with the following data in the range A1:D5:

	A	B	C	D
1	ga:year	ga:month	ga:channel	ga:sessions
2	2015	Jan	Search	46,936
3	2015	Jan	Email	922
4	2015	Jan	Referral	4,973
5	2015	Feb	Search	43,302

The formula bar shows the following formula: `=ArrayFormula(index(D2:D9,match($F2&$G2&H1,A2:A9&B2:B9&C2:C9,0)))`

The results are shown in the range F2:H5:

	F	G	H
	Year	Month	Search
2	2015	Jan	46,936
3	2015	Feb	43,302

I have a table of data in the range A1:D5, consisting of Google Analytics website data from early 2015. I want to lookup the data based on Year, Month and Channel, into cell H2.

I've setup a search table in the range F1:H3 (shown here next to the main table, but in reality they could be in different sheets) and my lookup formula goes into cells H2 and H3.

The Google Sheets formula is:

```
=ArrayFormula(index($D$2:$D$9,match($F2&$G2&H$1,$A$2:$A$9&$B$2:$B$9&$C$2:$C$9,0)))
```


And in Excel:

```
{=INDEX($D$2:$D$9,MATCH($F2&$G2&H$1,$A$2:$A$9&$B$2:$B$9&$C$2:$C$9,0))}
```

Let's break that down. The inner match formula is where the action happens:

First it concatenates my search criteria, the year, month and channel ($F2 \& G2 \& H1$) and then it searches across concatenated columns in the main table ($A2 : A9 \& B2 : B9 \& C2 : C9$). The Array Formula takes care of searching over the whole of each column.

For the example above, this diagram visualizes what's going on, showing the intermediate tables that are created on the fly for the lookup:



Note, you won't actually see these temporary search tables in your spreadsheet (I've shown them above for illustration only).

#11.

Use SUMIF to sum based on multiple criteria with OR logic

Use SUMIF() based on multiple criteria	
TOOL	Excel & Google Sheets
TYPE	Function
DIFFICULTY	Hard
OFFICIAL DOCUMENTATION	Excel docs / Google Sheets docs

How many products were sold by Company A or by Company B? I want to sum all of Company A's sales and add in all of Company B's sales. I could use two SUMIF() functions and add them, which is completely fine.

But that's not much fun, right?!? Let's create a really badass formula to solve the problem.

I have a table of data in columns A to C, with State location, a date and a count of products sold in that location. Now, say I wanted to count how many products were sold in the state of Texas. There are a few ways I could think of doing this: a filter, a pivot table, a subtotal or using the SUMIF() formula. The SUMIF formula takes the form:

```
=sumif(A2:A151, "Texas", C2:C151)
```

But what if I also wanted to see how many products were also sold in Iowa, in other words how many were sold in Texas **OR** Iowa? Well, as I mentioned, I could add together two SUMIF formulas but let's see if we can do with a single formula.

Google Sheets Method:

We use SUMIF but this time pass an array {"Texas", "Iowa"} as our criterion to be evaluated on. By making the formula an array formula, we get an array of answers back: {541, 421}. We then use the SUM() formula to finally add these together.

Here it is in all its glory:

```
=ArrayFormula(sum(sumif($A$2:$A$151,{"Texas","Iowa"},$C$2:$C$151)))
```

And here's a screenshot:

fx =ArrayFormula(sum(sumif(\$A\$2:\$A\$151,{"Texas","Iowa"},\$C\$2:\$C\$151)))										
	A	B	C	D	E	F	G	H	I	J
1	Location	Date	Count							
2	Maine	Dec 16, 2015	208		Texas	541				
3	Illinois	Dec 15, 2015	278		Iowa	421				
4	Montana	Dec 14, 2015	166			962				
5	Pennsylvania	Dec 13, 2015	130		Texas + Iowa	=ArrayFormula(sum(sumif(\$A\$2:\$A\$151,{"Texas","Iowa"},\$C\$2:\$C\$151)))				
6	Kansas	Dec 12, 2015	224							
7	Iowa	Dec 11, 2015	107							
8	Alabama	Dec 10, 2015	77							
9	North Dakota	Dec 9, 2015	225							
10	Alaska	Dec 8, 2015	174							
11	Missouri	Dec 7, 2015	204							
12	Vermont	Dec 6, 2015	270							
13	South Dakota	Dec 5, 2015	233							
14	Delaware	Dec 4, 2015	99							
15	New Jersey	Dec 3, 2015	107							
16	South Carolina	Dec 2, 2015	135							
17	Wisconsin	Dec 1, 2015	218							
18	Massachusetts	Nov 30, 2015	213							
19	Arizona	Nov 29, 2015	109							
20	Washington	Nov 28, 2015	149							

Excel Method:

Here's the equivalent formula in Microsoft Excel:

```
{=SUM(SUMIF($A$2:$A$151,{"Texas","Iowa"},$C$2:$C$151))}
```

The curly braces in Excel {} are equivalent to the ArrayFormula() in Google Sheets.

#12.

Use the FILTER() function on a range to return only the values you want

Using the FILTER() function	
TOOL	Google Sheets
TYPE	Function
DIFFICULTY	Medium
OFFICIAL DOCUMENTATION	Google Sheets docs

The FILTER function in Google Sheets is a powerful function that we can use to, well, *filter* our data.

Suppose you want to retrieve all values above a certain threshold? Or values that were greater than average? Or all even, or odd, values?

The FILTER function can easily do all of these, and more, with a single formula.

Suppose you have a list of values in column A and you want to extract specific values from the list, for example the numbers that are greater than average, or only the even numbers.

The FILTER function will return only the values that satisfy the conditions we've set. It takes two arguments, firstly the list of values we want to filter and secondly the conditions we're going to apply. The syntax is:

```
=FILTER("list of values", "conditions we're testing")
```

Some examples:

Filter for < 50	<code>=filter(A2:A20,A2:A20<50)</code>
Filter for > average	<code>=filter(A2:A20,A2:A20>AVERAGE(A2:A20))</code>
Filter for even values	<code>=ArrayFormula(filter(A2:A20,iseven(A2:A20)))</code>
Filter for odd values	<code>=ArrayFormula(filter(A2:A20,isodd(A2:A20)))</code>

Note that the last two of these formulas (the even and odd values) are **Array Formulas**. When you've typed in the FILTER formula, you'll need to hit **Ctrl + Shift + Enter** (on a PC) or **Cmd + Shift + Enter** (on a Mac), and Google will automatically add the `ArrayFormula()` wrapper.

	A	B	C	D	E
1		Filter for < 50	Filter for > average	Filter for even values	Filter for odd values
2	List of values	<code>=filter(A2:A20,A2:A20<50)</code>	<code>=filter(A2:A20,A2:A20>AVERAGE(A2:A20))</code>	<code>=ArrayFormula(filter(A2:A20,iseven(A2:A20)))</code>	<code>=ArrayFormula(filter(A2:A20,isodd(A2:A20)))</code>
3	220	24	416	220	189
4	416	14	418	416	203
5	189		251	418	251
6	418		331	24	331
7	203		311	190	311
8	251		399	422	219
9	331		422	84	59
10	311		355	14	399
11	219			144	73
12	59				355
13	24				
14	399				
15	190				
16	73				
17	422				
18	355				
19	84				
20	14				
21	144				
22					
23					

Here's another example using the FILTER function, to count and sum how many values lie between two dates.

	A	B	C	D	E	F
1	Dates	Values				Count
2	1/1/2015	1211		6/1/2015	6/30/2015	26
3	1/3/2015	1225				26
4	1/4/2015	1146				
5	1/5/2015	902				Sum
6	1/6/2015	1263				30480
7	1/8/2015	1293				
8	1/8/2015	1061				
9	1/10/2015	1136				
10	1/11/2015	1168				
11	1/11/2015	867				
12	1/12/2015	941				
13	1/12/2015	1031				
14	1/12/2015	937				

Specifically, I want to count how many values occur in June and then sum the total value in June, using the FILTER function.

Assume we have dates in column A, and values in column B, as shown in the spreadsheet image above.

To **count** the number of values in June, first put 6/1/15 in cell D2 and 6/30/15 in cell E2, and then try this formula in the next cell, as follows:

6/1/15	6/30/15	=count(filter(\$A\$2:\$A\$448,\$A\$2:\$A\$448<= \$E\$2,\$A\$2:\$A\$448>= \$D\$2))
--------	---------	---

Or in your Google sheet:

D	E	F	G	H	I	J
		Count				
6/1/2015	6/30/2015	=count(filter(\$A\$2:\$A\$448,\$A\$2:\$A\$448<= \$E\$2,\$A\$2:\$A\$448>= \$D\$2))				

But wait, there is a better way!

We can make our formula more concise, using the MONTH formula inside the filter:

```
=count(filter($A$2:$A$448,month($A$2:$A$448)=6))
```

What about summing the values in column B?

```
=SUM(filter($B$2:$B$448,month($A$2:$A$448)=6))
```

We've done two things, changed the filter range to column B and then wrapped the filter function in a SUM formula.

#13.

Use the RANK() function to find the order of a value in a list

Use the RANK() function to find order	
TOOL	Excel & Google Sheets
TYPE	Function
DIFFICULTY	Medium
OFFICIAL DOCUMENTATION	Excel docs / Google Sheets docs

The RANK() formula can be used to find the position, or rank, of a value in a list.

For example where would 76% rank in a list of exam scores? Let's take a look.

1. Assume we have a table setup with students and their exam results, as follows:

	A	B
1	Bill	76
2	Simon	54
3	Jess	65
4	Anna	81
5	Stephen	79
6	Lucy	60
7		
8		
9		

2. First, let's find out what rank the score of 76% was? Use the rank formula as follows:

```
=RANK ( 76 , B1 : B6 )
```

This gives us an answer of 3. In other words, 3rd out of 6 scores. That makes sense as we can see the scores in order are 81, 79, **76**, 65, 60 and 54.

3. What about finding out what rank Lucy came? Well, we need to start by using a VLOOKUP or INDEX/MATCH to find out what score Lucy got. Then we wrap that in the RANK formula above to get our answer.

```
=RANK ( VLOOKUP ( D1 , $A$1 : $B$6 , 2 , FALSE ) , $B$1 : $B$6 )
```

Or using INDEX/MATCH:

```
=RANK ( INDEX ( $B$1 : $B$6 , MATCH ( D1 , $A$1 : $A$6 , 0 ) ) , $B$1 : $B$6 )
```

Both of these give us the correct answer: 5. Lucy's result of 60% placed her 5th out of the 6 students:

	A	B	C	D	E	F	G
1	Bill	76		Lucy	5		
2	Simon	54					
3	Jess	65					
4	Anna	81					
5	Stephen	79					
6	Lucy	60					
7							
8							
9							

#14.

Retrieve only the unique values from a column


Use the UNIQUE() function to retrieve unique values	
TOOL	Google Sheets
TYPE	Function
DIFFICULTY	Easy
OFFICIAL DOCUMENTATION	Google Sheets docs

Google sheets has a very quick and easy function for retrieving a list of unique values from a column.

Simply use the =UNIQUE() function to get a list of unique values from the input range. Let's see this in action:

1. This example assumes you have a range of values in column A of your spreadsheet.
2. In an adjacent column, B in this example, enter the formula =UNIQUE(A1:A20) where A1 and A20 are the start and end cell references of your range.
3. The formula will then display a new range with just the unique values. Simple!

Here's a screenshot showing the UNIQUE function in action:



	A	B
1	17	=unique
2	18	UNIQUE
3	23	Unique rows in the provided
4	5	
5	43	
6	17	
7	89	
8	89	
9	18	
10	5	
11	5	
12	5	
13	5	

If this embedded video is not playing for you, [please check it out online here](#).

#15.

Writing code in your spreadsheets with the QUERY() function

Writing code with the QUERY() function	
TOOL	Google Sheets
TYPE	Function
DIFFICULTY	Hard
OFFICIAL DOCUMENTATION	Google Sheets docs

The QUERY() function in Google Sheets allows you to write code in your spreadsheets similar to Structured Query Language (SQL) that is the de-facto standard language for communicating with databases and querying datasets.

The QUERY function lets you run a query (kind of like a question) across a set of data and retrieve only results that answer that question.

It may seem abstract but it's a really useful way of importing and summarizing data in one fell swoop, especially as things get more complex.

Let's see how this function works with two examples:

Example 1: Importing and Filtering Simpsons data

First step is to perform some basic web scraping (see tip 22) to import some Simpsons data into my Google Sheet:

```
=IMPORTHTML("https://en.wikipedia.org/wiki/The_Simpsons", "table", 3)
```

Now what if we quickly wanted to find out which seasons had more than 25 million viewers? We can do that easily with the QUERY function. The QUERY function takes 3 arguments:

1. The data, which in this case is the table created by the function above.
2. The query (question) which uses a SQL like code to specify what you want returned e.g. “select Col1, Col4, Col8 where Col8 > 25” - select columns 1, 4 and 8 when column 8 is greater than 25. The format has to be “Col1” for Column 1, “Col2” for Column 2 etc.
3. When the data has headers, as ours does, put a 1 into our formula for the last argument.

The complete formula is:

```
=QUERY (IMPORTHTML ("https://en.wikipedia.org/wiki/The_Simpsons", "table", 3), "select Col1, Col4, Col8 where Col8 > 25", 1)
```

The result is a smaller table showing only the relevant data we requested:

	A	B	C	D	E	F	G	H
1	Season	TV season	Viewers (millions)					
2		1 1989–90	33.5					
3		2 1990–91	33.6					
4		3 1991–92	25.5					
5		4 1992–93	28.6					
6								
7								

Example 2: Quickly creating summary statistics

I’ve set up this quick, fictitious table showing purchases of electronic products on particular days:

Date	Company	Purchases
9-Nov-2015	Apple	7
11-Nov-2015	Samsung	8
14-Nov-2015	Xaomi	5
14-Nov-2015	Blackberry	8

14-Nov-2015	Apple	2
18-Nov-2015	Apple	2
20-Nov-2015	Samsung	1
21-Nov-2015	Xaomi	9
22-Nov-2015	Xaomi	10
25-Nov-2015	Xaomi	5
29-Nov-2015	Apple	10

What if we quickly want to summarize this data and find out the total, average, max, min etc. per product, like this:

Company	Total purchases	Average purchases/day	Minimum purchases/day	Maximum purchases/day
Xaomi	29	7.25	5	10
Apple	21	5.25	2	10
Samsung	9	4.5	1	8
Blackberry	8	8	8	8

Sure we could do it using a table with lots of formulas but there's a quicker way - a single, albeit, complex QUERY function.

Basically, we write a single line of code inside the QUERY function to do all the aggregation for us:

```
=QUERY(A8:C19,"select B, sum(C), avg(C), min(C), max(C) group by B
order by sum(C) desc label sum(C) 'Total purchases', avg(C) 'Average
purchases', min(C) 'Minimum purchases', max(C) 'Maximum purchases'",1)
```

#16.

Database functions in your spreadsheets

Use your spreadsheet like a database	
TOOL	Excel & Google Sheets
TYPE	Function
DIFFICULTY	Medium / Hard
OFFICIAL DOCUMENTATION	Excel docs / Google Sheets docs

Excel and Google Sheets both feature a set of database functions. These work by treating a range of data in your spreadsheet as if it were a table in a database, allowing you to perform calculations on that data based on criteria. Easy huh?

The reason these functions are useful is their ability to specify conditions by which to summarize your data. Let's see that in action, using Excel.

Imagine you have this menu of sandwiches and subs in your restaurant:

Key	Subs/Sandwiches	Category	Preparation	Calories
1	Cheese sandwich	Vegetarian	5	650
2	Meatball sub		15	1100
3	Philly Cheesesteak sub		20	1100
4	Tuna Melt		15	750
5	Veggie Melt	Vegetarian	14	550
6	Nutella sandwich	Vegetarian	3	440
7	PBJ sandwich	Vegetarian	2	400
8	Chicken Parmigiana		30	1000

How many subs take less than 25 minutes to prepare but still pack greater than 700 calories?

It wouldn't take too long to just figure that out manually, but what if your table had a few thousand rows long...good luck with that.

We could potentially do this with some of the standard Excel functions, but we can also do it quickly and easily with a single database function, by following these steps:

1. Setup a mini-table with containing your filter conditions:

Key	Subs/Sandwiches	Category	Preparation	Calories
	*sub		<25	>700

Note we've entered **sub* which uses the wildcard (*) so that we retrieve all results where the Subs/Sandwiches column ends with the word "sub".

2. Enter the following database function DCOUNT()

```
=DCOUNT (A1:E9, "Key", A12:E13)
```

where A1:E9 is our main "database" table and A12:E13 is our filter conditions table.

3. Hit enter and the DCOUNT() function will count the number of items in the Key column where the criteria are all satisfied. So you'll see the answer to our original question: **2**.

There are 2 subs that take less than 25 minutes to prepare but still pack in more than 700 artery-thickening calories! I'll take the Meatball sub please!

See this in action in the following screenshot:

DCOUNTA *fx* =DCOUNT(A1:E9,"Key",A12:E13)

	A	B	C	D	E	F
1	Key	Recipe	Category	Preparation	Calories	
2	1	Cheese sandwich	Vegetarian	5	650	
3	2	Meatball sub		15	1100	
4	3	Philly Cheesesteak sub		20	1100	
5	4	Tuna Melt		15	750	
6	5	Veggie Melt	Vegetarian	14	550	
7	6	Nutella sandwich	Vegetarian	3	440	
8	7	PBJ sandwich	Vegetarian	2	400	
9	8	Chicken Parmigiana		30	1000	
10						
11						
12	Key	Recipe	Category	Preparation	Calories	
13		*sub		<25	>700	
14						
15						
16						
17	=DCOUNT(A1:E9,"Key",A12:E13)					
18						
19						
20						

#17.

Create charts with the REPT() function

In cell charts with the REPT() function	
TOOL	Excel & Google Sheets
TYPE	Function
DIFFICULTY	Medium
OFFICIAL DOCUMENTATION	Excel docs / Google Sheets docs


Here's a simple way to create a mini chart using the REPT() function, which just repeats a text string for a given number of times.

We simply repeat the letter "I" inside the cell for a specific number of times to represent a bar, as follows:

```
=REPT("I", C2)
```

Feel free to try using other characters instead of I. Making the font bold and red adds emphasis.

Check out the following example, showing the number of wins for each team in the English Premier League:



If this embedded video is not playing for you, [please check it out online here.](#)

#18.

Sparklines

Sparklines	
TOOL	Excel & Google Sheets
TYPE	Function / Chart
DIFFICULTY	Medium
OFFICIAL DOCUMENTATION	Excel docs / Google Sheets docs


Sparklines are small, simple charts without axes, which exist inside a single cell. They were first created by statistician and data visualization legend [Edward Tufte](#).

They're a wonderful, quick way for visually showing a result, without needing the complexity of a full-blown chart, and they work well for datasets based on a timescale.

In Excel, sparklines are added by going to the Insert menu and choosing Sparkline. From there, you can specify a line or bar chart (or others) and then specify the range of cells containing data, as well as the cell where you want the sparkline to show.

There are many other options I encourage you to explore, such as color coding the smallest and largest values, or the first and last values in your data.

The following GIF shows how to create sparklines in Excel, as well as some of the additional features:



If this embedded video is not playing for you, [please check it out online here.](#)

In Google Sheets, things are a little different.

Use the formula:

```
=sparkline(data, [options])
```

where data refers to a range of values to plot the sparkline. The optional options argument is used to specify things like chart type (line, bar, column or winloss), color and other specific settings.

#19.

Motion charts in Google Sheets

Creating motion charts in Google Sheets	
TOOL	Google Sheets
TYPE	Function
DIFFICULTY	Medium
OFFICIAL DOCUMENTATION	Google Sheets docs

Ok, be patient with this one. It's very cool, but it's rather buggy and can be frustrating to use. It's based on Flash technology so won't work everywhere and it may not even survive in its current form for much longer, with the [much publicized demise of Flash](#).

So why they heck am I showing you this? Because it's impressive what a few clicks can create, so I wanted to show you an example of an animated chart. For an example of a motion chart being used very effectively check out minute 4 of this TED talk (watching the entire talk is highly recommended):

http://www.ted.com/talks/hans_rosling_shows_the_best_stats_you_ve_ever_seen#t-228887

For the motion chart to work, the data needs to be formatted in a very particular way, as follows:


The first column should contain entities (e.g. countries), the second is time (e.g. years), followed by 2-4 numeric or string columns.

For this example, I grabbed some population and GDP data to plot (I make no promises to this being a valid analysis, these were purely used for charting purposes!):

Country	Year	Pre School Population	GDP per capita
---------	------	-----------------------	----------------


Mali	1999	1396194	273.7
Mali	2000	1430221	240.4
Mali	2001	1467336	265.3
Mali	2002	1519242	272.0
Mali	2003	1571371	349.2
...

The table also included data for Angola, Tanzania and United Kingdom, for the years 1999 to 2013. Once we have data in a format like this, then click **Insert > Chart** and then select **Motion Chart** as shown in the following image:



You can adjust the colors and add labels to the bubbles in your chart. Be warned though, these charts have a habit of resetting to the default blue, so I'd wait until you've finished all other work on your sheet, and then add these flourishes as a final step.

The resulting motion chart is:



If this embedded video is not playing for you, [please check it out online here.](#)


#20.

Exploring data in Google Sheets with the EXPLORE button

Exploring data in Google Sheets with the EXPLORE button	
TOOL	Google Sheets
TYPE	Method
DIFFICULTY	Easy / Medium
OFFICIAL DOCUMENTATION	Google Sheets docs

Sometimes the hardest part of reporting your results can be deciding on suitable charts to visualize your data. You've got a nice, neat table of data but you're not sure where to begin with the charts and you've got better things to do with your time than click and create hundreds of charts until you find the right one.

Well, thanks to a recent release of Google Sheets, the computer can do all the hard work for you. In the bottom right corner of your Google Sheet you'll see a green Explore button. Highlight your data, then go ahead and click on that Explore button:



Hey presto! Like magic, Google opens a pane on the right of your screen where it has *automagically* created a bunch of summary statistics and charts for you to....well, *explore*.

Scroll down the pane to see the many different charts Google has created. Now, they're not all going to be suitable or meaningful, but it's a good place to start.

If you want to use a particular chart, then you can either drag the chart into your sheet or click the Insert Chart button, as shown in the following screenshot:

The screenshot shows a Google Sheet with columns I, J, and K. Column I contains values from 1,499,834 to 8,130,731. Column J contains values from 1,812,059 to 13,277,777. Column K contains values from 2 to 21. The 'Explore' pane is open on the right, displaying summary statistics: SUM (2951149335), AVERAGE (8504753.128), MIN (0.015), MAX (216531488), and COUNTA (348). Below the statistics is a line chart with the x-axis labeled 'Rate/Year' ranging from 3.00% to 15.00% and the y-axis ranging from 0 to 180,000. A tooltip 'Insert chart' is visible over the chart area.

I	J	K
1,499,834	1,812,059	2
1,647,846	2,022,077	2
1,813,862	2,262,038	2
2,000,220	2,536,508	3
2,209,567	2,850,765	3
2,444,899	3,210,910	4
2,709,609	3,623,993	4
3,007,541	4,098,168	5
3,343,043	4,642,859	6
3,721,041	5,268,957	7
4,147,106	5,989,053	8
4,627,548	6,817,696	9
5,169,504	7,771,696	11
5,781,050	8,870,476	13
6,471,323	10,136,473	15
7,250,654	11,595,600	18
8,130,731	13,277,777	21

The official Youtube video introducing the [Explore feature can be viewed here](#).

#21.

Creating histograms with the frequency function

Create histograms with FREQUENCY() and Array Formulas	
TOOL	Excel & Google Sheets
TYPE	Function
DIFFICULTY	Medium / Hard
OFFICIAL DOCUMENTATION	Excel docs / Google Sheets docs

Let's look at analyzing data using the FREQUENCY() formula to see the distribution of our data, perhaps to try and determine if our data is normally distributed. We use the FREQUENCY() formula to categorize our data into groups, or bins, that we specify and then count how many values occur in each of these bins.

Let's see this in action with an example:

1. Assume we have a dataset of 100 values, between 1 and 100, in column A.
2. In column C, we set up our bins as follows:

Bins
1 to 10
11 to 20
21 to 30
31 to 40
41 to 50
51 to 60
61 to 70

71 to 80
81 to 90
91 to 100

We want to count how many of the original values lie within each of these bins. How many values in our dataset lie between 1 and 10? Between 11 and 20? Between 21 and 30? Etc.

- Next, in column D, select the upper bound of each bin. We need to do this step to use in our frequency formula. So our table in columns C and D now looks like this:

Bins	Upper bound
1 to 10	10
11 to 20	20
21 to 30	30
31 to 40	40
41 to 50	50
51 to 60	60
61 to 70	70
71 to 80	80
81 to 90	90
91 to 100	100

- Finally we're ready to use the FREQUENCY() formula. Starting from cell E2, select the range E2 to E11, i.e. the blank cells adjacent to the upper bounds we just created.
- Next, type in this formula but don't hit enter just yet:

```
=FREQUENCY(A2:A101, D2:D11)
```

Now things diverge a little, depending on whether you're in Excel or Google Sheets:


- In Excel, hit **Ctrl + Shift + Enter** on a PC, or **Cmd + Shift + Enter** on a Mac to create your Array formula and fill in the entire column with the correct frequency calculations for each bin count.

`{=FREQUENCY (A2:A101, D2:D11) }`

- In Google Sheets, hit **Ctrl + Shift + Enter** on a PC, or **Cmd + Shift + Enter** on a Mac to create the array formula and then hit **Enter** again to fully populate the column. The formula created in Google Sheets takes this form:

`=ArrayFormula (FREQUENCY (A2:A101, D2:D11))`

Let's see that in action in this short video:



If this embedded video is not playing for you, [please check it out online here.](#)

P.S. the 5-second method in Google Sheets:

Now then, will you be mad if I tell you that Google Sheets has a histogram chart that can create that same chart for us in about 5 seconds flat?

Highlight your whole range of data, go to the **Insert > Chart** menu and select the **Histogram** option. Notice how Google has already created the bins for us and counted all of the values in each one!

Same data as above, charted in 5-seconds flat, count:

The screenshot shows a Google Sheets spreadsheet with a dataset in column A. The data points are: 87, 46, 63, 2, 9, 1, 10, 76, 75, 15, 37, 14, 97, 91, 3, 2, 94, 94, 50, 38, 25, 47, 4, 79, 70. The 'Insert' menu is open, and the 'Chart...' option is highlighted. A play button icon is overlaid on the bottom left of the spreadsheet area.

Row	Value
77	87
78	46
79	63
80	2
81	9
82	1
83	10
84	76
85	75
86	15
87	37
88	14
89	97
90	91
91	3
92	2
93	94
94	94
95	50
96	38
97	25
98	47
99	4
100	79
101	70

If this embedded video is not playing for you, [please check it out online here.](#)

#22.

Web scraping with the IMPORTHTML() and IMPORTXML() functions

Web scraping using Google Sheets	
TOOL	Google Sheets
TYPE	Function
DIFFICULTY	Medium
OFFICIAL DOCUMENTATION	Google Sheets docs

Ever wanted to access data on a website in your spreadsheets, so you can do some quantitative analysis? Copying and pasting is a pain! Thankfully Google Sheets have a family of import functions that can do the job for us.

In a blank workbook, try copying the following formula into cell A1 (as shown in the screenshot below):

```
=IMPORTHTML("https://en.wikipedia.org/wiki/List_of_highest-grossing_films","table",1)
```

This imports the table directly into our Google Sheet, as follows:

The screenshot shows the Google Sheets interface with the formula bar containing the following formula:

```
=importhtml("https://en.wikipedia.org/wiki/List_of_highest-grossing_films","table",1)
```

The resulting table is displayed below the formula bar:

	A	B	C	D	E	F
1	Rank	Peak	Title	Worldwide gross	Year	Reference(s)
2		1	1 *Avatar*	\$2,787,965,087	2009	[# 1][# 2]
3		2	1 *Titanic*	\$2,186,772,302	1997	[# 3][# 4]
4		3	3 *Jurassic World*	\$1,668,891,572	2015	[# 5][# 6]
5		4	3 *The Avengers*	\$1,519,557,910	2012	[# 7][# 8]
6		5	4 *Furious 7*	\$1,514,827,481	2015	[# 9][# 10]
7		6	5 *Avengers: Age of Ultron*	\$1,405,035,767	2015	[# 11][# 10]
8		7	3 *Harry Potter and the Chamber of Secrets*	\$1,341,511,219	2011	[# 12][# 13]

There are different flavors of import functions available, allowing you to access different specific details on web pages.

Let's see another example of web scraping. We want to find out the author of a specific article on a website. I had to do this for a client recently where they had a huge long list of blog posts and wanted to know the author of each post. It would have been super tedious to extract manually, so this automatic method was a real time saver.

Try this example by typing this web address into cell A1:

<http://www.nytimes.com/2015/09/23/us/los-angeles-plans-100-million-effort-to-end-homelessness.html>

and then add the following formula to cell B1:

```
=IMPORTXML(A1, "//*[@class='byline-author']")
```

This accesses the web address in cell A1 and pulls out the author's name for that article.

The output of this example is:

	A	B
1	http://www.nytimes.com/2015/09/23/us/los-angeles-plans-100-million-effort-to-end-homelessness.html	JENNIFER MEDINA
2		

I've written a couple of blog posts about web scraping with more information and examples:

<http://www.benlcollins.com/spreadsheets/google-sheet-web-scraper/>

<http://www.benlcollins.com/spreadsheets/import-social-media-data/>

#23.


Tag people in comments

Tag people in comments	
TOOL	Google Sheets
TYPE	Function
DIFFICULTY	Easy
OFFICIAL DOCUMENTATION	Google Sheets docs

You can alert people to specific comments in your Google Sheets by adding an email notification to the comment.

You do this very easily by typing a **+** or **@** and then the email address you want to notify. As you start typing, Google will auto-suggest email addresses from your contacts list. The person will receive an email alert to notify them about the comment mention.

Note: this does NOT grant that person any editing rights over the document!

	A	B	C	D
1	Profit (m)	Ben Collins		
2	\$0.87			any D
3	\$0.70			
4	\$1.79	Check this out +benlcollins@		
5	\$1.36	"Ben Collins" <benlcollins@gmail.com>		
6	\$2.50	"Ben Collins" <benlcollins@hotmail.com>		
7				
8				

#24.

Detect the language of a given string of text

Detecting languages in Google Sheets	
TOOL	Google Sheets
TYPE	Function
DIFFICULTY	Easy
OFFICIAL DOCUMENTATION	Google Sheets docs

Did you know that Google Sheets has a built-in language detector? You can use a simple formula to quickly find out what language a text is written in.

Assuming you have some text of unknown language in cell A1, simply add this formula into the adjacent cell:

```
=DETECTLANGUAGE(A1)
```

And here's a screenshot showing it working for "hello" in a few different languages:

	A	B	C	D
1	buenos días	es	Spanish	
2	bonjour	fr	French	
3	guten tag	de	German	
4	hello	en	English	
5	Здравствуйте	ru	Russian	
6	こんにちは	ja	Japanese	
7	您好	zh	Chinese	
8				
9				

Note: the formula returns a two letter code of the language it detects, so you'll need to convert to full country code, as shown in column C above.

See this Wikipedia article for a full list of two letter language codes:

https://en.wikipedia.org/wiki/List_of_ISO_639-1_codes

#25.

Translating languages in Google spreadsheets

Build a language translation tool in Google Sheets

TOOL	Google Sheets
TYPE	Function
DIFFICULTY	Medium
OFFICIAL DOCUMENTATION	Google Sheets docs

So we talked about detecting languages in your Google Sheets in tip 24, but what about translating text from one language to another? Yes, it's possible in Google sheets with the GOOGLETRANSLATE() function! And better still, you can combine it with the DETECTLANGUAGE() function to create a powerful tool for detecting and translating languages in your Google Sheets.

The formula takes 3 arguments - the text to translate, the language the text is currently in, and finally, the language you want the text translated into.

Check it out, in its standard form:

	A	B	C
1	Hello World	Bonjour le monde	
2			
3			

fx | =GOOGLETRANSLATE(A1,"en","fr")

Now, check it out combined with the DETECTLANGUAGE function to give a powerful language translation tool in your Google sheets (in this example, the original text, cell A1, is in Russian, which Google detects):

<i>fx</i> =GOOGLETRANSLATE(A1, DETECTLANGUAGE(A1), "EN")			
	A	B	
1	Привет мир	Hello World	
2			
3			

#26.

Check for a valid email in Google Sheets

Check for valid emails with the ISEMAIL() function	
TOOL	Google Sheets
TYPE	Function
DIFFICULTY	Easy
OFFICIAL DOCUMENTATION	Google Sheets docs

Use the ISEMAIL function in Google Sheets to determine if an email in your spreadsheet is valid. The formula takes one argument (either a string of text or a cell reference) and checks whether it's a valid email address by returning a TRUE or FALSE. For example:

```
=ISEMAIL("joe.bloggs@example.com")
```

```
=ISEMAIL(A1)
```

fx		=isemail(A1)	
	A	B	
1	ben@example.com	TRUE	
2	ben@example	FALSE	
3	ben example.com	FALSE	
4	ben@@example.com	FALSE	
5	ben@example.	FALSE	
6			

#27.

Check for a valid URL (web address) in Google Sheets

Check for a valid web address with ISURL() function	
TOOL	Google Sheets
TYPE	Function
DIFFICULTY	Easy
OFFICIAL DOCUMENTATION	Google Sheets docs

Quickly and easily validate URLs in your spreadsheets with the ISURL() function. The formula takes one argument (the value representing a URL) and returns a TRUE or FALSE depending on whether the URL is valid or not. For example:

```
=ISURL("http://www.benlcollins.com/")
```

```
=ISURL("www.benlcollins.com/")
```

```
=ISURL("benlcollins.com/")
```


	A	B
1	http://www.benlcollins.com/	TRUE
2	www.benlcollins.com/	TRUE
3	benlcollins.com/	TRUE
4	www.benlcollins	FALSE
5	www.benlcollins.c	FALSE
6		
7		

#28.

Sourcing financial data in seconds with the GOOGLFINANCE() function

Sourcing financial data in seconds with GOOGLFINANCE() function	
TOOL	Google Sheets
TYPE	Function
DIFFICULTY	Medium
OFFICIAL DOCUMENTATION	Google Sheets docs

I bet you I can draw a chart of Facebook's stock price over the past year in less than 30 seconds, without leaving my Google spreadsheet. Don't believe me? Check this out:



If this embedded video is not playing for you, [please check it out online here.](#)

The GOOGLFINANCE() function fetches financial data from the Google Finance portal.

It has a whole host of different parameters (e.g. the ticker, the dates, daily v. weekly, price v. volume, etc...) so definitely worth exploring the [official docs](#).


The formula is:

```
=GOOGLFINANCE("FB", "price", today()-365, today(), "daily")
```

So, how does it work? The formula takes five arguments:

1. The ticker of the company you want financial info for, e.g. "FB", "Goog"
2. The attribute you're interested in, e.g. "price", "marketcap"
3. The start date of the period you're interested in
4. The end date of the period you're interested in (here I've used the today() formula)
5. The interval, either "daily" or "weekly"

Here is the formula in action, complete with a line chart:



#29.

Getting data from different Google Sheets

Getting data from a different Google Sheet using IMPORTRANGE() function

TOOL	Google Sheets
TYPE	Function
DIFFICULTY	Hard
OFFICIAL DOCUMENTATION	Google Sheets docs

If you're coming from an Excel background, you may be surprised to hear that it's considerably more difficult to link to a different workbook in Google Sheets. You can't simply click on a cell or range in another workbook and expect the data to pull across.

Instead, you need to use the IMPORTRANGE() function to actually import the data you're after.

Here's how:

1. Make sure your current workbook is open (Workbook A) and then open the other workbook you want to connect to (Workbook B).
2. Copy the second part of the URL from the workbook you want to connect to (Workbook B). Copy **EVERYTHING THAT COMES AFTER** the “/d/” section in the web address, to your clipboard. The following diagram shows this reference highlighted in red (I've deliberately obscured my reference):



3. In your current workbook, Workbook A, enter the following formula and paste in your reference:

```
=IMPORTRANGE("[.....your ref....."],
```


4. Next specify the name of the sheet in the other workbook, Workbook B, that you want to get data from, in this case "Sheet1". Our formula now:

```
=IMPORTRANGE("[.....your ref....."],"Sheet1!
```

5. Then specify the range you want to import. This could be a single cell, or as in this example, a range. Our formula now:

```
=IMPORTRANGE("[.....your ref....."],"Sheet1!A1:D9")
```

6. You'll then see the following error message. Click "Allow access" to connect your sheets:



7. Finally, hit enter and you should see the data you're after be pulled into your new sheet!

Told you it was harder than Excel!!

#30.

Project future results with the GROWTH() function

Project future results with the GROWTH() function	
TOOL	Excel & Google Sheets
TYPE	Function
DIFFICULTY	Medium
OFFICIAL DOCUMENTATION	Excel docs / Google Sheets docs

The GROWTH() function in Excel and Google Sheets can be used to predict future values. Given a set of partial data about a trend, the GROWTH() function can be used to extrapolate that trend and predict future values.


Consider this table for growth of some metric over the first 6 months of the year:

1	100
2	105
3	119
4	123
5	157
6	169

Then, using the GROWTH function:

```
=growth(B1:B6,A1:A6,A8:A13)
```

We can extrapolate for months 7 - 12 to estimate data for those months, as shown in the following screenshot:



I've include a chart to show the exponential curve based off all 12 values.

#31.

Adding new lines into formulas to make them more readable

Make formulas more readable	
TOOL	Excel & Google Sheets
TYPE	Function
DIFFICULTY	Easy
OFFICIAL DOCUMENTATION	N/a

Ever used a long and complex formula and struggled to see all the different components in one long line? Well, this might help you out; we can insert carriage returns into our formulas to make them more readable.

In Excel and Google Sheets, we simply use Alt + Return to achieve this! Easy as pie.

Our **Before** example:

```
=IF(( SUMIF(A2:A20, "EXPIRED", B2:B20) ) > 2000, "More than $2000  
Expired", "Less than $200 Expired")
```

Our **After** example:

```
=IF(  
( SUMIF(A2:A20, "EXPIRED", B2:B20) ) > 2000,  
"More than $2000 Expired",  
"Less than $200 Expired"  
)
```

	A	B	C	D	E
1	STATUS	List of values			
2	NOTPAID	220)	
3	NOTPAID	416			
4	EXPIRED	189			
5	EXPIRED	418			
6	APPROVED	203			
7	APPROVED	251			
8	EXPIRED	331			
9	EXPIRED	311			
10	ISSUED	219			
11	ISSUED	59			
12	APPROVED	24			
13	APPROVED	399			
14	APPROVED	190			
15	EXPIRED	73			
16	EXPIRED	422			
17	ISSUED	355			
18	EXPIRED	84			
19	NOTPAID	14			
20	REJECTED	144			
21					

```

IF
( SUMIF(A2:A20,
"EXPIRED",
B2:B20)
) > 2000,
"More than $2000 Expired",
"Less than $200 Expired"
)

```

For example, consider the complex query formula from tip 15 - difficult to read in one long line, but with strategic carriage returns added, it's much more readable.

Compare before:

```

=QUERY(A8:C19,"select B, sum(C), avg(C), min(C), max(C) group by B
order by sum(C) desc label sum(C) 'Total purchases', avg(C) 'Average
purchases', min(C) 'Minimum purchases', max(C) 'Maximum purchases'",1)

```

To after:

```

=query(A8:C19,
"select B, sum(C), avg(C), min(C), max(C)
group by B
order by sum(C) desc
label sum(C) 'Total purchases',
      avg(C) 'Average purchases',
      min(C) 'Minimum purchases',
      max(C) 'Maximum purchases'",

```

1)

#32.

Use the CHAR() function to get funky characters into your worksheets

Inserting funky characters into your worksheets	
TOOL	Excel & Google Sheets
TYPE	Function
DIFFICULTY	Easy
OFFICIAL DOCUMENTATION	Excel docs / Google Sheets docs

Ever needed to insert the copyright symbol ©, the trademark symbol ™, or the registered trademark symbol ®, but not been able to find them? Or what about funky math symbols like infinity ∞ or the number pi π?

Use the CHAR() function to achieve this easily, provided you know the relevant number of the symbol you're interested in (this can be the challenge!). The CHAR function converts a number into a character according to the current Unicode table.

Symbol	Excel	Google Sheets
©	=char(169)	=char(169)
®	=char(168)	=char(174)
™	=char(170)	n/k
¶	=char(166)	=char(182)
π	=char(185)	=char(960)
∞	=char(176)	n/k

Handy hint to find what you're looking for:

- Type `=char(row())` into cell A1 and drag down column A of your spreadsheet. This works up to row 255 for Excel but into the thousands for Google Sheets, because it includes all the different language character sets. Then you can search through to find interesting symbols.

A few fun examples to finish:

- If you're on a Mac, try `=char(240)`
- In Google Sheets, try `=char(936)` or `=char(1161)`

This site is a good place to search for interesting characters: [Graphemica](#)

#33.

Use CHOOSE and a random number generator to pick from a list

Using CHOOSE and a random number generator to pick from a list	
TOOL	Excel & Google Sheets
TYPE	Function
DIFFICULTY	Medium
OFFICIAL DOCUMENTATION	Excel docs / Google Sheets docs

This is a cool and reasonably straightforward technique to pick randomly from a defined list, using the CHOOSE() and RANDBETWEEN() functions.

Suppose you want to create some fake data to test your formulas or application.

In this example, we want fictitious place names in our dataset, but we want to restrict to choosing from just New York, Washington DC, San Francisco, Seattle or Denver, say. We can use this technique to randomly and repeatedly pick from that list.

Here's the formula:

```
=CHOOSE (RANDBETWEEN (1, 5), "New York", "Washington DC", "San Francisco", "Seattle", "Denver")
```

We then copy this formula down a column to create as many iterations as required.

So what's going on under the hood?

We start with the RANDBETWEEN (1, 5) sub-formula. This generates a random integer between 1 and 5 each time.

Then CHOOSE is used to pick from the list of entries ("New York", "Washington DC", "San Francisco", "Seattle", "Denver").

Each time the formula is used, a new random number is generated and a new pick made from the list.

One point to note using this technique is to ensure the bounds of your random number generator (5) match to the number of items in your list (5).

#34.

Excel intersect operator

Excel intersect operator	
TOOL	Excel
TYPE	Function
DIFFICULTY	Medium
OFFICIAL DOCUMENTATION	N/a

Consider the following table of sales data by region and by quarter:

	Q1	Q2	Q3	Q4
North	287	354	424	395
South	303	495	279	346
East	331	282	269	394
West	406	329	319	274

Nothing fancy here, just a simple sales table!

Now, what if we want to know the Sales in Q3 for the East region? Easy, we'd just read it off as 269.

But we can also use the intersect method in Excel to do that. Yes, overkill here but you'll see why it can be quite powerful shortly.

Take a look at this screenshot:

	A	B	C	D	E	F	G	H	I
1		Q1	Q2	Q3	Q4				
2	North	287	354	424	395		East Sales in Q3:	=4:4 D:D	
3	South	303	495	279	346				
4	East	331	282	269	394				
5	West	406	329	319	274				
6									
7									
8									

Typing in the formula:

=4:4 D:D

with a space between the 4:4 and D:D selects the intersection of these two ranges, in this case D4, with value 269.

The advantage of this method is that it allows us to do some computations on the intersected ranges, for example, what is the sum of sales totals for the South region in Q1 and Q2?

SUM									
	A	B	C	D	E	F	G	H	I
1		Q1	Q2	Q3	Q4				
2	North	287	354	424	395		East Sales in Q3:	269	
3	South	303	495	279	346		South Q1 and Q2:	=SUM(3:3 B:C)	
4	East	331	282	269	394				
5	West	406	329	319	274				
6									
7									
8									

The formula uses a sum to add the intersections together, i.e South Q1 + South Q2:

=sum(3:3 B:C)

Note, for non-contiguous ranges in your intersect, you'll have to combine using the sum formula in the regular way, i.e. with a comma, as follows:

	A	B	C	D	E	F	G	H	I	J
1		Q1	Q2	Q3	Q4					
2	North	287	354	424	395		East Sales in Q3:	269		
3	South	303	495	279	346		South Q1 and Q4:	=SUM(3:3 B:B,3:3 E:E)		
4	East	331	282	269	394			SUM(number1, [number2], [num		
5	West	406	329	319	274					
6										
7										
8										

Here the formula is:

=sum(3:3 B:B,3:3 E:E)

Remember those distance calculation tables in the back of old road atlases. This union operator is perfect for creating one of those in Excel.

#35.

Bonus links to other weird and wonderful uses of spreadsheets

Create award winning art in Excel:

<http://www.amusingplanet.com/2014/01/tatsuo-horiuchi-creates-amazing-art.html>

Generate a sudoku puzzle and solver:

<http://sites.mcpher.com/share/Home/excelquirks/recursionlink/sudoku-generator-and-solver>

Build a map in Excel using a pivot table:

<https://www.linkedin.com/pulse/excel-map-hack-john-nelson>

Comprehensive selection of advanced (and crazy!) Excel charts:

<https://sites.google.com/site/e90e50charts/>

3D Dancing pendulums in Excel:

<http://chandoo.org/wp/2011/07/06/3d-dancing-pendulums/>

For the coders amongst you, here's a couple of articles on how to use Google Sheets as a back-end database for your web application and how to connect to a Javascript front-end for charting:

<http://kovalent.co/blog/google-docs-as-a-backend/>

&

<https://davidwalsh.name/js-charting-google-sheets>

Know anymore? Share them at benjcollins@gmail.com and I'll feature you here with a credit!

Closing Comments.

Thanks for downloading and reading this ebook! It was great fun to put together so I hope you've enjoyed it and found it helpful.

Corrections, comments or suggestions are encouraged. Let me know what you think. You can reach me directly at benllcollins@gmail.com

Got any weird and wacky spreadsheet tips or tricks? I plan to expand this guide and will credit anyone who contributes.

Remember, keep summing, sorting, filtering, pivoting, charting and analyzing, but above all keep having fun!

Thanks!

Ben Collins

January 2016

Contact details

Website: <http://www.benllcollins.com/>
Linkedin: <https://www.linkedin.com/in/benllcollins>
Twitter: <https://twitter.com/benllcollins>
Github: <https://github.com/benllcollins>